```
Clear[G633];
G633[thermo_, p_, t_, odmod_:"equilibrium" (*for o-d e-m, specify "ordered" or "disordered" *)]:=

   Module[{t0 = 298.15, p0 = 0.001, vkt, tc0,q20, tc,q2, pth, q, vterm, gbase, sfterm, comp,
           ta, tb, theta, R = 0.0083144, god, sod, sfds,
           name, enthalpy, entropy, volume, cpa, cpb, cpc, cpd, alpha0, kappa0, kappa0p, kappa0pp,
           dkappa0dT, aqcp, landaut, smax, vmax,  sfdh, sfdhv, sfw, sfwv, sfn, sffac, isliq, cpterms, vv},

      {name, (* natoms, comp, *) enthalpy, entropy, volume,
             cpa, cpb, cpc, cpd, alpha0, kappa0, kappa0p, kappa0pp, theta, dkappa0dT,
             landaut, smax, vmax, aqcp, sfdh, sfdhv, sfw, sfwv, sfn, sffac} = thermo;

      isliq = StringTake[name, -1] == "L";

      cpterms = cpa (t - t0) + cpb (t^2 - t0^2)/2 - cpc (1/t - 1/t0) + 2 cpd (t^(1/2) - t0^(1/2)) -
                t (2 cpa (Log[t^(1/2)] - Log[t0^(1/2)]) + cpb (t - t0) - cpc/2 (t^-2 - t0^-2) - 2 cpd (t^(-1/2) - t0^(-1/2)));

      pth := If[isliq, 0, theta alpha0 kappa0 / (Exp[theta/t0] * (theta/t0)^2 / (Exp[theta/t0] - 1)^2) *
                                                                  (1/(Exp[theta/t] - 1) - 1/(Exp[theta/t0] - 1) )];

      if[isliq, kappa0 += dkappa0dT (t - t0)];

      ta = (1 + kappa0p)/(1 + kappa0p + kappa0 * kappa0pp);
           tb = (kappa0p + kappa0p^2 - kappa0 * kappa0pp)/(kappa0 * (1 + kappa0p));
           tc = (1 + kappa0p + kappa0 * kappa0pp)/(kappa0p + kappa0p^2 - kappa0 * kappa0pp);

      vv = volume If[isliq, Exp[alpha0 (t - t0)], 1];

      vterm = vv * ((p - p0) (1 - ta) + ta (-(1 + tb * (p  - pth))^(1 - tc) +
                                           (1 + tb * (p0 - pth))^(1 - tc))/(tb (tc - 1)))/
                                  ((1 - ta) +   ta (1 + tb * p0)^(-tc));  (* new (5-6-19) *)

      gbase = N[enthalpy - t entropy + cpterms + vterm];

      god =
        Which[sfn > 0,  (* order-disorder by sf *)

               Which[
                   odmod==="ordered",  0,

                   odmod==="disordered",
                     sfds = If[sffac<0,
                               sffac * R * (Log[1/(sfn+1)] + sfn*Log[sfn/(sfn+1)])*(1/sffac-sfn)/(sfn+1),
                               sffac * R * (Log[1/(sfn+1)] + sfn*Log[sfn/(sfn+1)]) ];
                     sfdh + sfds t + sfdhv p,

                   odmod==="equilibrium",
                     If[sffac<0,           (* ihe[] solves for qq by interval-halving *)
                         q = ihe[ sfdh + p sfdhv + (sfw + p sfwv) (2 qq - 1) +
                                    sfn/(sfn + 1) R t * (Log[sfn (1 - qq)/(1 + sfn qq)] -
                                                         sffac Log[(1 - qq)/(sfn + qq)]),
                                  qq, {}, 10^-8, 1 - 10^-8, 30];

                         sod = ((1 + sfn q) Log[(1 + sfn q)/(sfn+1)] + sfn (1-q) Log[sfn*(1-q)/(sfn+1)] -
                                sffac (sfn (1-q) Log[(1-q)/(sfn+1)] + sfn (sfn+q) Log[(sfn+q)/(sfn+1)] ))/(sfn+1),

                         q = ihe[ sfdh + p sfdhv + (sfw + p sfwv) (2 qq - 1) +
                                    sffac sfn/(sfn + 1) R t * Log[sfn (1 - qq)^2/((1 + sfn qq) (sfn + qq))],
                                  qq, {}, 10^-8, 1 - 10^-8, 30];

                         sod = sffac ((1+sfn q) Log[(1 + sfn q)/(sfn + 1)] +
                                       sfn (1 - q) Log[(1 - q)/(sfn + 1)] +
                                                    sfn (1 - q) Log[sfn (1 - q)/(sfn + 1)] +
                                       sfn (sfn + q) Log[(sfn + q)/(sfn + 1)]) / (sfn + 1)];

                         sfdh + p sfdhv + q (sfw - sfdh + p (sfwv - sfdhv)) -
                                  q^2 (sfw + p sfwv) + R t sod ],

               smax > 0,  (* order-disorder by landau *)
                  tc0 = landaut;
                  q20 = Sqrt[1 - t0 / tc0];

                  Which[odmod==="ordered",
                            smax tc0 (-2/3 + q20 (1 - q20^2/3)) - t smax (q20 - 1) + p vmax (q20 - 1),

                        odmod==="disordered",
                            smax tc0 q20 (1 - q20^2/3) - t smax q20 + p vmax q20,

                        odmod==="equilibrium",
                            tc =  tc0 + If[vmax == 0, 0, p vmax / smax];
                            q2 =  If[t > tc,       0, Sqrt[(tc - t) / tc0]];
                            smax (tc0 (q20 (1 - 1/3 q20^2) + 1/3 q2^3) - q2 tc) - t smax (q20 - q2) + p vmax q20 ],

               True,   0 ];

           N[gbase + god] ];
```